

# Gnumeric: электронная таблица для всех

И.А.Хахаев, © 2007-2010

## 9 Моделирование рисков методом Монте-Карло

В состав Gnumeric входит средство для моделирования по методу Монте-Карло. При использовании метода Монте-Карло используются выборки случайных чисел для решения задач, в которых течение времени не играет существенной роли. Другими словами, на очередную выборку не влияют предыдущие выборки.

При моделировании по методу Монте-Карло используются функции генерации случайных чисел, а результаты представляются вместе со статистическими характеристиками, которые можно затем анализировать.

В данной главе рассмотрен пример из официального руководства по Gnumeric, причём исправлены некоторые ошибки и неточности.

### 9.1 *Общее описание задачи*

Одна из классических расчётных задач – задача о продавцах газет. Продавцы покупают газеты за 33 цента каждую и продают по 50 центов. Непроданные газеты идут на макулатуру по 5 центов за штуку. Газеты продаются распространителям пачками по 10 штук. Спрос на газеты может быть поделён на «замечательный», «нормальный» и «плохой» с вероятностями 0.35, 0.45 и 0.20 соответственно, причём текущий спрос не зависит от предыдущего дня. Задача продавца – определить оптимальное количество газет в ситуации, когда спрос не вполне известен, то есть добиться устойчивого дохода.

Уравнение для определения дневного дохода для продавца выглядит следующим образом:

$$\text{Доход} = [(\text{Выручка}) - (\text{Себестоимость}) + (\text{Макулатура})]$$

Остаётся добавить, что количество закупленных от поставщика газет может изменяться от 40 до 100 включительно, а количество проданных газет также кратно 10.

### 9.2 *Построение модели*

для построения модели в Gnumeric будем использовать два листа - лист «Доход» для вычисления дохода и лист «Таблицы спроса» для таблиц, требуемых для модельных наборов данных, задающих параметры спроса.

На листе «Доход» создадим таблицу расчёта дохода, как показано на рис. 9.1.

	А	В	С	Д	Е
9	Вычисление дохода				
10					
11	<b>Переменные</b>	выручка	себестоимость	макулатура	доход
12	<b>Значение</b>	25	16,5	0	
13	<b>Коэффициент</b>	0,5	0,33	0,05	
14					
15	<b>Данные</b>	Количество			
16	<b>Закупка</b>	50			

Рисунок 9.1. Исходная таблица вычисления дохода

Таблицу для вычисления дохода начнём с девятой строки. У нас есть три переменные – выручка от продаж, себестоимость газет и стоимость макулатуры, для которых на каждую единицу товара заданы коэффициенты 0.5, 0.33 и 0.05 соответственно. Запишем эти коэффициенты в ячейки от B13 до D13. В ячейках от B12 до D12 запишем формулы для дохода от продаж, себестоимости и стоимости макулатуры, как показано в таблице 1. В ячейке E12 запишем формулу для вычисления прибыли.

Таблица 1

Адрес ячейки	Значение или формула
B12	=B13*min(B16;B20)
C12	=C13*B16
D12	=D13*max(0;B16-B20)
E12	=B12-C12+D12
B13	0,5
C13	0,33
D13	0,05
B16	50

Нужно заметить, что на этом этапе в некоторых ячейках появятся сообщения «N/A!» («нет данных»). Как только модель будет построена полностью, эти сообщения исчезнут.

В ячейке B20 будет задаваться случайное количество проданных газет («спрос»). Поскольку нельзя продать больше, чем закуплено у поставщика, выручка определяется количеством проданных газет, если закуплено больше, чем продано и ограничивается количеством закупленных газет, если спрос превышает это количество (функция min( ) при расчёте выручки).

Формула в ячейке D12 означает, что в макулатуру можно сдать только непроданные газеты, поэтому если всё продано (а также если спрос превышает количество закупленных газет), то количество макулатуры будет 0.

Начальное количество закупленных газет установим в 50.

Далее на листе «Таблицы спроса» сформируем модельные параметры спроса в соответствии с рис. 9.2, 9.3 и 9.3.

Термин «вероятность» в рассматриваемом примере означает значение функции плотности распределения, а «интегральная вероятность» – значение функции распределения.

	А	В	С	Д	Е
1					
2					
3	<b>Итого дня</b>	Вероятность	Интегральная вероятность		
4	<b>Замечательно</b>	0,35	0,35		
5	<b>Нормально</b>	0,45	0,8		
6	<b>Плохо</b>	0,2	1		
7					

Рисунок 9.2. Распределение уровней спроса

	А	В	С	Д
9				
10	<b>Распределение ежедневного спроса на газеты</b>			
11	<b>Спрос</b>	Замечательно	Нормально	Плохо
12	40	0,03	0,1	0,44
13	50	0,05	0,18	0,22
14	60	0,15	0,4	0,16
15	70	0,2	0,2	0,12
16	80	0,35	0,08	0,06
17	90	0,15	0,04	0
18	100	0,07	0	0
19				

Рисунок 9.3. Распределение спроса в зависимости от количества газет

	А	В	С	Д	Е	Ф	Г
19							
20	<b>Функции распределения спроса</b>						
21		Интегральная вероятность			Значения		
22	<b>Спрос</b>	Замечательно	Нормально	Плохо	Замечательно	Нормальнс	Плохо
23	40	0,03	0,1	0,44	0	0	0
24	50	0,08	0,28	0,66	0,03	0,1	0,44
25	60	0,23	0,68	0,82	0,08	0,28	0,66
26	70	0,43	0,88	0,94	0,23	0,68	0,82
27	80	0,78	0,92	1	0,43	0,88	0,94
28	90	0,93	1		0,78	0,92	1
29	100	1			0,93	1	
30							

Рисунок 9.4. Функции распределения спроса

На рис. 9.2 показаны плотность распределения и значения функции распределения уровней спроса, на рис 9.3 – функции плотности распределения вероятности продажи заданного количества газет для каждого из вариантов спроса, а на рис. 9.4 – вспомогательные функции распределения вероятностей спроса.

Дополнительными допущениями является то, что 40 газет будут проданы при любых условиях, а вот 100 – только при наиболее благоприятных обстоятельствах.

Теперь снова перейдём на лист «Доход» и продолжим ввод формул, нужных для работы модели. Адреса ячеек и соответствующие формулы показаны в таблице 2.

Таблица 2

Адрес ячейки	Значение или формула
B17	=rand()
C17	=if(B17<'Таблицы спроса'!C4;"Замечательно";if(B17<'Таблицы спроса'!C5;"Нормально";"Плохо"))
B18	=rand()
B20	=lookup(C17;\$B\$23:\$D\$23;\$B\$24:\$D\$24)
B21	=E12
B23	Замечательно
C23	Нормально
D23	Плохо
B24	=lookup(B18;'Таблицы спроса'!\$E\$23:\$E\$29;'Таблицы спроса'!\$A\$23:\$A\$29)
C24	=lookup(B18;'Таблицы спроса'!\$F\$23:\$F\$29;'Таблицы спроса'!\$A\$23:\$A\$29)
D24	=lookup(B18;'Таблицы спроса'!\$G\$23:\$G\$29;'Таблицы спроса'!\$A\$23:\$A\$29)

Случайное число в ячейке B17 определяет уровень (состояние) спроса, который выводится в ячейку C17. Случайное число в ячейке B18 определяет количества проданных газет для каждого уровня спроса (ячейки B24, C24 и D24). В соответствии с ранее заданным в C17 уровнем спроса в ячейке B20 получаем текущий спрос на газеты, из которого уже рассчитывается доход.

Итоговый вариант листа «Доход» должен выглядеть примерно так, как показано на рис. 9.5.

	А	В	С	Д	Е
9	Вычисление дохода				
10					
11	<b>Переменные</b>	выручка	себестоимость	макулатура	доход
12	<b>Значение</b>	25	16,5	0	8,5
13	<b>Коэффициент</b>	0,5	0,33	0,05	
14					
15	<b>Данные</b>	Количество			
16	<b>Закупка</b>	50			
17	<b>Итог дня</b>	0,6873233079327	Нормально		
18		0,3227079860445			
19	<b>Результат</b>				
20	<b>Спрос</b>	60			
21	<b>Доход</b>	8,5			
22					
23		Замечательно	Нормально	Плохо	
24		70	60	40	
25					

Рисунок 9.5. Итоговый лист для вычисления дохода

Нажимая на клавишу F9 («Правка/Пересчитать» в главном меню) можем наблюдать за изменением чисел и, соответственно, за изменением дохода.

### 9.3 Использование модели

Понятно, что наблюдать за изменениями чисел в ячейках ЭТ при пересчёте очень увлекательно, но не очень продуктивно. Инструмент моделирования в Gnumeric («Сервис/Моделирование...» в главном меню) позволяет автоматически проделать большое число пересчётов и проанализировать получаемые результаты.

При вызове инструмента моделирования появляется диалог «Моделирование рисков», как обычно содержащий несколько вкладок.

Вкладка «Переменные» (рис. 9.6) обеспечивает определение диапазонов входных и выходных данных. Для рассматриваемого примера эти диапазоны должны быть определены в соответствии с рис. 9.6.




Рисунок 9.6. Определение диапазонов для входных и выходных данных

На вкладке «Параметры» (рис. 9.7) настраивается режим вычислений – определяется количество вариантов модели («Rounds»), количество пересчётов (итераций) для каждого варианта и предел времени, в течение которого проводятся вычисления (в секундах). Предел времени нужен для предотвращения слишком длительной загрузки вычислительной системы. Если модель очень сложна или составлена некорректно, вычисления прекратятся по достижении этого предела времени.

Использование вариантов модели («Rounds») будет рассмотрено позднее.




Рисунок 9.7. Настройки вычислений

Вкладка «Вывод» (рис. 9.8) является традиционной для Gnumeric и позволяет задать расположение результатов работы программы. В нашем случае целесообразно использовать вариант «Новый лист».






Рисунок 9.8. Настройка расположения результатов моделирования

После нажатия на кнопку «OK» в диалоге «Моделирование рисков» диалоговое окно не закрывается, и на вкладке «Итог» можно увидеть, что появились какие-то результаты (рис. 9.9).




Рисунок 9.9. Результаты моделирования в диалоговом окне

В области «Сводка результатов:» вкладки «Итог» можно увидеть значения переменных, полученные в результате выполнения указанного в области «Итог моделирования» количества итераций. Изучать эти значения в области вкладки диалогового окна не очень удобно, поэтому имеет смысл закрыть диалог и посмотреть результаты на появившемся в книге ЭТ листе «Отчёт о моделировании» (рис. 10).

	A	B	C	D	E
1		<b>gnumeric Моделирование рисков 1.10.5 Отчет</b>			
2		<b>Книга: [file:///mnt/win_d/gnumeric-book/simulation.gnumeric]Доход</b>			
3		<b>Отчёт создан:Sun Oct 10 14:33:05 2010</b>			
4					
5					
6		<b>ИТОГО</b>			
7			<b>Мин</b>	<b>Среднее</b>	<b>Макс</b>
8		<b>Спрос Количество</b>	40	64,34	100
9		<b>Доход Количество</b>	4	7,825	8,5
10		<b>(Ввод) Закупка Количество</b>	50	50	50
11		<b>(Ввод) Итог дня Количество</b>	5,702740021134E-05	0,50254637421886	0,9995027511702
12		<b>(Ввод) Количество</b>	0,00012427875973	0,50664222257099	0,9991930672988

Рисунок 9.10. Фрагмент листа с отчётом о моделировании

В правой части этого листа имеется ещё несколько статистических параметров, полученных в результате вычислений. Их можно получить, проделав все описанные выше действия, а назначение этих параметров (на английском языке) описано в оригинальном руководстве по Gnumeric. поэтому здесь они не приводятся.

Полученный результат можно интерпретировать следующим образом: «Если продавец будет покупать 50 газет, то его доход будет варьироваться от 4 до 8,5 у.е. и в среднем будет составлять 7,825 у.е. Доход будет не менее 4 у.е. в самых неблагоприятных условиях, но не более 8,5 у.е. – в самых благоприятных».

Теперь можно проделать аналогичные действия, установив количество закупаемых от поставщика газет (ячейка B16 на листе «Доход») в 50, 60 и т. д. Однако процесс перебора количества закупаемых газет тоже можно автоматизировать. Для этого в Gnumeric имеется функция SIMTABLE(), находящаяся в категории «Случайные числа».

Для того, чтобы просчитать модель с различными количествами закупаемых газет, запишем в ячейку B12 формулу =simtable(50;60;70;80;90) (крайние случаи с количествами 40 и 100 брать не будем).

Поскольку в этом случае имеется 5 вариантов модели, то на вкладке настройки режима вычислений диалога «Моделирование рисков» требуется изменить значение «Last round #:» в соответствии с количеством аргументов функции SIMTABLE() (рис. 9.11).

Следует заметить, что номера вариантов совпадают с номерами аргументов функции SIMTABLE(). Поэтому, если хочется посмотреть, что будет при количествах закупленных газет от 60 до 80, то в качестве параметра «First round #:» нужно поставить 2, а для «Last round #:» – 4.




Рисунок 9.11. Настройка вычислений для нескольких вариантов модели

После нажатия на кнопку «OK» на вкладке «Итог» окажутся активными кнопки «Next Sim.» и «Prev. Sim» (рис. 9.12), которые позволяют в области «Сводка результатов» видеть результаты по каждому варианту модели (при каждом значении количества закупленных газет из набора значений, определённых в качестве аргументов функции `SIMTABLE( )`).




Рисунок 9.12. Просмотр результатов по вариантам модели

Полная сводка результатов отображается на листе «Отчёт о моделировании» (рис. 9.13), на котором поместился только фрагмент сводки результатов, поэтому основные результаты приведены в таблице 3.

	A	B	C	D	E
6	SUMMARY OF SIMULATION ROUND #1				
7			<b>Мин</b>	<b>Среднее</b>	<b>Макс</b>
8	<b>Спрос</b>	<b>Количество</b>	40	64,41	100
9	<b>Доход</b>	<b>Количество</b>	4	7,8205	8,5
10	<b>(Ввод)</b>	<b>Закупка Количество</b>	50	50	50
11	<b>(Ввод)</b>	<b>Итог дня Количество</b>	0,003352172062	0,498789753087	0,999636536864
12	<b>(Ввод)</b>	<b>Количество</b>	0,000578942701	0,499960621012	0,998962741086
13					
14					
15	SUMMARY OF SIMULATION ROUND #2				
16			<b>Мин</b>	<b>Среднее</b>	<b>Макс</b>
17	<b>Спрос</b>	<b>Количество</b>	40	64,68	100
18	<b>Доход</b>	<b>Количество</b>	1,2	8,2515	10,2
19	<b>(Ввод)</b>	<b>Закупка Количество</b>	60	60	60
20	<b>(Ввод)</b>	<b>Итог дня Количество</b>	0,000345595932	0,489356043302	0,998794125497
21	<b>(Ввод)</b>	<b>Количество</b>	0,001553605354	0,497441035969	0,999672653996
22					
23					
24	SUMMARY OF SIMULATION ROUND #3				
25			<b>Мин</b>	<b>Среднее</b>	<b>Макс</b>
26	<b>Спрос</b>	<b>Количество</b>	40	63,9	100
27	<b>Доход</b>	<b>Количество</b>	-1,6	7,364	11,9
28	<b>(Ввод)</b>	<b>Закупка Количество</b>	70	70	70
29	<b>(Ввод)</b>	<b>Итог дня Количество</b>	0,000806831029	0,500545577640	0,999709810391
30	<b>(Ввод)</b>	<b>Количество</b>	0,001793650310	0,495337517129	0,999865479445
31					
32					
33	SUMMARY OF SIMULATION ROUND #4				
34			<b>Мин</b>	<b>Среднее</b>	<b>Макс</b>
35	<b>Спрос</b>	<b>Количество</b>	40	65,19	100
36	<b>Доход</b>	<b>Количество</b>	-4,4	6,2065	13,6

Рисунок 9.13. Результаты моделирования по вариантам входных данных

Из таблицы 3 видно, что закупка 60 газет даёт наиболее надёжный доход – средний доход максимален и никогда не получается убытка.

При увеличении количества закупаемых газет растёт возможный максимальный доход, однако средний доход уменьшается и возрастает риск убытков.

Очевидно, что такой результат получается при параметрах модели, заданных на листе «Таблицы спроса». При других вероятностях и распределениях вероятностей результаты также будут отличаться.

Таблица 3: Краткий итог моделирования

Закуплено газет	Доход, у.е.		
	Минимальный	Средний	Максимальный
50	4	7,82	8,5
60	1,2	8,25	10,2
70	-1,6	7,36	11,9
80	-4,4	6,21	13,6
90	-7,2	3,91	15,3

В заключение этой главы хотелось бы заметить, что программные средства для экономического моделирования стоят тысячи долларов США даже при условии их использования в образовательных целях (а для использования в коммерческих целях – значительно дороже). Поэтому использование Gnumeric (да и вообще свободного программного обеспечения) и для такого рода задач может оказаться привлекательным.